Computational Logic

Pure (Declarative) Logic Programs




Pure Logic Programs (Overview)

e Programs that only make use of unification.

e They are completely logical: the set of computed answers is exactly the set of
logical consequences.

o Computed answers: all calls that compute successfully

e Allow to program declaratively: declare the problem
(specifications as programs)

e They have full computational power.

1. Database programming.

2. Arithmetics.

3. Data structure manipulation.
4. Recursive programming.




Database Programming

e A Logic Database is a set of facts and rules (i.e., a logic program):

father_of (john,peter) <- . grandfather_of (L,M) <- father_of(L,N),
father_of (john,mary) <- . father_of (N,M).
father_of (peter,michael) <- grandfather_of (X,Y) <- father_of (X,Z),
mother_of (Z,Y) .

mother_of (mary, david) <- .

e Given such database, a logic programming system can answer questions
(queries) such as:

<- father_of (john, peter). <- grandfather_of (X, michael).
Answer: Yes Answer: {X = john}

<- father of (john, david). <- grandfather of (X, Y).
Answer: No Answer: {X = john,Y = michael}
<- father of (john, X). Answer: {X = john,Y = david}
Answer: {X = peter} <- grandfather of (X, X).
Answer: {X = mary} Answer: No

e Rules for grandmother of (X, Y)?




Database Programming (Contd.)

e Another example:

Power
nl " resistor(power,nl) <- .
0o, 312 resistor (power,n2) <- .
' - 22
—O— transistor(n2,ground,nl) <- .
n4 n5 transistor(n3,n4,n2) <- .
- —0— _
13 transistor (nb,ground,nd) <- .

inverter (Input,Output) <-
transistor (Input,ground,Output), resistor(power,Output) .
nand_gate (Inputl, Input2,0utput) <-
transistor (Inputl,X,0utput), transistor(Input2,ground,X),
resistor(power,Qutput) .
and_gate(Inputl, Input2,Output) <-
nand_gate(Inputl,Input2,X), inverter (X, Output).

e Query and gate(Ini,In2,0ut) has solution: {In1=n3, In2=n5, Out=nl}




Structured Data and Data Abstraction

e The circuit example revisited:
resistor(rl,power,nl) <- . transistor(tl,n2,ground,nl) <- .
resistor (r2,power,n2) <- . transistor(t2,n3,n4,n2) <- .
transistor(t3,n5,ground,nd) <- .

inverter(inv(T,R),Input,Output) <-
transistor (T, Input,ground,Output), resistor(R,power,Qutput) .

nand_gate (nand(T1,T2,R),Inputl, Input2,0utput) <-
transistor(T1,Inputl,X,0utput), transistor(T2,Input2,ground,X),
resistor(R,power,Qutput) .

and_gate(and(N,I),Inputl, Input2,0utput) <-
nand_gate (N, Inputl,Input2,X), inverter(I,X,0utput).

e The query |<- and_gate(G,Inl,In2,0ut).

has solution: |{G=and (nand (t2,t3,r2) ,inv(tl,r1)),In1=n3,In2=n5,0ut=n1}




Logic Programs and the Relational DB Model

Traditional — Codd’s Relational Model

File Relation Table
Record Tuple Row
Field Attribute Column
e Example:
Name Town Years
Name | Age | Sex Brown | London 15
Brown| 20 | M Brown York 5
Jones | 21 F Jones Paris 21
Smith| 36 | M Smith | Brussels 15
Smith | Santander 5
Person
Lived—in

e The order of the rows is immaterial.

e (Duplicate rows are not allowed)




Logic Programs and the Relational DB Model (Contd.)

Relational Database — Logic Programming
Relation Name

— Predicate symbol
Relation — Procedure consisting of ground facts
(facts without variables)
Tuple — Ground fact
Attribute — Argument of predicate
e Example: Name | Age | Sex
person(brown,20,male) <- . Brown | 20 | M
: Jones | 21 F
person(jones,21,female) <- . smith T 36 | M
person(smith,36,male) <- .

e Example: Name | Town | Years
lived_in(brown,london,15) <- . Brown | London 15
lived in(brown,york,5) <- . Brown |  York 5
lived in(jones,paris,21) <- . Jones Paris 21

, , , Smith | Brussels 15
lived_in(smith,brussels,15) <- . Smith | Santander 5
lived_in(smith,santander,5) <- .




Logic Programs and the Relational DB Model (Contd.)

e The operations of the relational model are easily implemented as rules.

< Union:
runion_s(Xy,...,X,) «— r(Xy,...,X,).
runion_s(Xy,...,X,) «— s(Xq,...,X,).
o Set Difference:
rdiff s(X,,...,X,) «— r(Xy,...,X,,), not s(Xy,...,X,).
rdiff s(Xy,...,X,) «— s(Xy,...,X,,), not r(Xq,...,X,).
(we postpone the discussion on negation until later.)
< Cartesian Product:
rXs(Xq,. ., X0, Xnmtts s Xpman) «— v(Xq,.. ., X)) ,s(X a1, s Xogn)
o Projection:
r13(X;,X3) «— r(X;,X5,X3).
o Selection:
r_selected (X7, X5, X3) «— r(Xi,Xs,X3),<(X,,X3).
(see later for definition of </2)




Logic Programs and the Relational DB Model (Contd.)

e Derived operations — some can be expressed more directly in LP:

o Intersection:
rmeet_s(X;y,...,X,) «— r(Xy,...,X,), s(Xy,...,X,).
& Join:
r joinX2 s(Xi,...,X,) «— r(Xy,Xy,X5,..., X)), s(X{,Xo,X5,..., X)),

e Duplicates an issue: see “setof” later in Prolog.




Deductive Databases

e The subject of “deductive databases” uses these ideas to develop logic-based
databases.

o Often syntactic restrictions (a subset of definite programs) used
(e.g. “Datalog” — no functors, no existential variables).

o Variations of a “bottom-up” execution strategy used: Use the T}, operator
(explained in the theory part) to compute the model, restrict to the query.




Recursive Programming

e Example: ancestors.
parent (X,Y) <- father(X,Y).
parent (X,Y) <- mother(X,Y).

ancestor(X,Y) <- parent(X,Y).

ancestor(X,Y) <- parent(X,Z), parent(Z,Y).

ancestor(X,Y) <- parent(X,Z), parent(Z,W), parent(W,Y).
ancestor(X,Y) <- parent(X,Z), parent(Z,W), parent(W,K), parent(X,Y).

e Defining ancestor recursively:
parent (X,Y) <- father(X,Y).
parent (X,Y) <- mother(X,Y).

ancestor(X,Y) <- parent(X,Y).
ancestor(X,Y) <- parent(X,Z), ancestor(Z,Y).

e Exercise: define “related”, “cousin”, “same generation”, etc.




Types

e Type: a (possibly infinite) set of terms.
e Type definition: A program defining a type.
e Example: Weekday:

o Set of terms to represent: Monday, Tuesday, Wednesday, . ..

o Type definition:
is weekday(’Monday’) <- .
is weekday(’Tuesday’) <- .

e Example: Date (weekday * day in the month):

o Set of terms to represent: date(’Monday’,23), date(Tuesday,24), ...

o Type definition:
is date(date(W,D)) <- is weekday(W), is day of month(D).
is_day_of month(1) <- .
is_day_of month(2) <- .

is_day_of month(31) <- .




Recursive Programming: Recursive Types

e Recursive types: defined by recursive logic programs.
e Example: natural numbers (simplest recursive data type):

o Set of terms to represent: 0, s(0), s(s(0)), ...

o Type definition:
nat(0) <- .
nat(s(X)) <- natX).

A minimal recursive predicate:
one unit clause and one recursive clause (with a single body literal).

e We can reason about complexity, for a given class of queries (“mode”).
E.g., for mode nat (ground) complexity is linear in size of number.

e Example: integers:

o Set of terms to represent: 0, s(0), -s(0),...
o Type definition:

integer( X) <- nat(X).

integer(-X) <- nat(X).




Recursive Programming: Arithmetic

e Defining the natural order (<) of natural numbers:

less_or_equal(0,X) <- nat(X).
less_or_equal(s(X),s(Y)) <- less_or_equal(X,Y).

e Multiple uses: less_or_equal(s(0),s(s(0))), less_or_equal(X,0),...
e Multiple solutions: less_or_equal (X,s(0)), less_or_equal(s(s(0)),Y), etc.

e Addition:
plus(0,X,X) <- nat(X).
plus(s(X),Y,s(Z2)) <- plus(X,Y,Z).

e Multiple uses: plus(s(s(0)),s(0),Z), plus(s(s(0)),Y,s(0))

e Multiple solutions: plus(X,Y,s(s(s(0)))), etc.




Recursive Programming: Arithmetic (Contd.)

e Another possible definition of addition:
plus(X,0,X) <- nat(X).
plus(X,s(Y),s(Z2)) <- plus(X,Y,Z).

e The meaning of plus is the same if both definitions are combined.

e Not recommended: several proof trees for the same query — not efficient, not
concise. We look for minimal axiomatizations.

e The art of logic programming: finding compact and computationally efficient
formulations!

o Try to define: times(X,Y,Z) (Z = X*Y), exp(N,X,Y) (Y = X%),
factorial (N,F) (F = N!), minimum(N1,N2,Min), ...




Recursive Programming: Arithmetic (Contd.)

e Definition of mod (X,Y,Z)
“Z Is the remainder from dividing X by Y”

FOQst.X=Y*Q+Zand Z <Y):
mod (X,Y,Z) <- less(Z, Y), times(Y,Q,W), plus(W,Z,X).

less(0,s(X)) <- nat(X).
less(s(X),s(Y)) <- less(X,Y).
e Another possible definition:
mod (X,Y,X) <- less(X, Y).
mod(X,Y,Z) <- plus(X1,Y,X), mod(X1,Y,Z).

e The second is much more efficient than the first one
(compare the size of the proof trees).




Recursive Programming: Arithmetic/Functions

e The Ackermann function:
ackermann(0O,N) = N+1
ackermann(M,0) = ackermann(M-1,1)
ackermann(M,N) = ackermann(M-1,ackermann(M,N-1))

e In Peano arithmetic:
ackermann (0O,N) = s(N)
ackermann(s(M),0) = ackermann(M,s(0))
ackermann(s(M),s(N)) = ackermann(M,ackermann(s(M),N))

e Can be defined as:
ackermann(0O,N,s(N)) <- .
ackermann(s(M),0,Val) <- ackermann(M,s(0),Val).
ackermann(s(M),s(N),Val) <- ackermann(s(M),N,Vall),
ackermann(M,Vall,Val).
e In general, functions can be coded as a predicate with one more argument, which
represents the output (and additional syntactic sugar often available).

e Syntactic support available (see, e.g., the Ciao functions package).




Recursive Programming: Lists

e Type definition (no syntactic sugar):
list([1) <- .
list(.(X,Y)) <- list(Y).

e Type definition (with syntactic sugar):
list([]) <- .
list ([X]Y]) <- list(Y).

e List concatenation (e.g., a list traversal):
append([],Ys,Ys) <- .
append ([X|Y],Ys, [X|Zs]) <- append(Xs,Ys,Zs).




Recursive Programming: Binary Trees

e Represented by a ternary functor tree (Element,Left,Right).
e Empty tree represented by void.
e Definition:
binary_tree(void) <- .
binary_tree(tree(Element,Left,Right)) <-

binary_tree(Left),
binary_tree(Right) .

e Defining tree_member (Element,Tree):

tree_member (X,tree(X,Left,Right)) <- .
tree_member (X,tree(Y,Left,Right)) <- tree_member (X,Left).
tree_member (X,tree(Y,Left,Right)) <- tree_member (X,Right).




Recursive Programming: Binary Trees (Contd.)

e Defining pre_order(Tree,Order):

pre_order(void, []) <- .

pre_order (tree(X,Left,Right) ,Order) <-
pre_order (Left,OrderLeft),
pre_order (Right,0OrderRight),
append ( [X|OrderLeft] ,0rderRight,Order) .

e Define in_order (Tree,Order), post_order(Tree,Order).




e In Prolog:

Creating a Binary Tree in Pascal and LP

T = tree(3, tree(2,void,void), tree(5,void,void)) 2

t"right“data :

e In Pascal. void void void void
new(t) ;
type tree = “treerec; new(t~left);
treerec = record new(t right);
data : integer; t"left " left := nil;
left : tree; t"left"right := nil;
right: tree; t right"left := nil,;
end; t right"right := nil,;
t"data := 3;
var t : tree; t"left"data :=




Polymorphism

e Note that the two definitions of member/2 can be used simultaneously:

1t_member (X, [X]Y]) <- list(Y).
1t_member(X,[_|T]) <- 1lt_member(X,T).

1t_member (X,tree(X,L,R)) <- binary_tree(L), binary_tree(R).
1t_member (X,tree(Y,L,R)) <- binary_tree(R), lt_member(X,L).
1t_member (X,tree(Y,L,R)) <- binary_tree(L), lt_member(X,R).

Lists only unify with the first two clauses, trees with clauses 3-5!

e <— 1t member (X, [b,a,c]).
X=b; X=a; X=¢c

e <- 1t member(X,tree(b,tree(a,void,void) ,tree(c,void,void))).
X=Db; X=a; X=c¢

e Also, try (somewat surprising): <- 1t_member(M,T).




Recursive Programming: Manipulating Symbolic Expressions

e Recognizing polynomials in some term X:

o X is a polynomial in X
¢ a constant is a polynomial in X
© sums, differences and products of polynomials in X are polynomials

¢ also polynomials raised to the power of a natural number and the quotient of a
polynomial by a constant

polynomial (X,X) <- .

polynomial (Term,X) <- pconstant(Term) .

polynomial (Term1+Term2,X) <- polynomial (Terml,X), polynomial(Term2,X).
polynomial (Term1-Term2,X) <- polynomial(Terml,X), polynomial(Term2,X).
polynomial (Term1*Term2,X) <- polynomial (Terml,X), polynomial(Term2,X).
polynomial (Term1/Term2,X) <- polynomial (Terml,X), pconstant(Term2).

polynomial (Term1~N,X) <- polynomial (Terml,X), nat(N).




Recursive Programming: Manipulating Symb. Expressions (Contd.)

e Symbolic differentiation: deriv(Expression, X, DifferentiatedExpression)

deriv(X,X,s(0)) <- .

deriv(C,X,0) <- pconstant(C).
deriv(U+V,X,DU+DV) <- deriv(U,X,DU), deriv(V,X,DV).
deriv(U-V,X,DU-DV) <- deriv(U,X,DU), deriv(V,X,DV).
deriv (U*V,X,DUxV+U*DV) <- deriv(U,X,DU), deriv(V,X,DV).
deriv(U/V,X, (DUxV-U*DV) /V~s(s(0))) <- deriv(U,X,DU), deriv(V,X,DV).
deriv(U~s(N),X,s(N)*xU"N*xDU) <- deriv(U,X,DU), nat(N).
deriv(log(U) ,X,DU/U) <- deriv(U,X,DU).

e <- deriv(s(s(s(0)))*x+s(s(0)),x,Y).

e A simplification step can be added.




Recursive Programming: Graphs

e Usual: make use of another data structure, e.g., lists
o Graphs as lists of edges.
e Alternative: make use of Prolog’s program database
o Declare the graph using facts in the program.
edge(a,b) <- .

edge(b,c) <- .
edge(c,a) <- . edge(d,a) <- .

e Paths in a graph: path(X,Y) iff there is a path in the graph from node X to node Y.

path(A,B) <- edge(A,B).
path(A,B) <- edge(A,X), path(X,B).

e Circuit: a closed path. circuit iff there is a path in the graph from a node to itself.

circuit <- path(A,A).




Recursive Programming: Graphs (Exercises)

e Modify circuit/O so that it gives the circuit.
(You have to modify also path/2)

e Propose a solution for handling several graphs in our representation.
e Propose a suitable representation of graphs as data structures.

e Define the previous predicates for your representation.

e Consider unconnected graphs (there is a subset of nodes not connected in any
way to the rest) versus connected graphs.

e Consider directed versus undirected graphs.

e Try path(a,d). Solve the problem.




Recursive Programming: Automata (Graphs)

e Recognizing the sequence of characters accepted by the following
non-deterministic, finite automaton (NDFA):

a
where O is both the initial
@ @ )b and the final state.
b

e Strings are represented as lists of constants (e.g., [a,b,b]).

e Program:
initial(q0) <- . delta(q0,a,ql) <- .
delta(ql,b,q0) <- .
final(q0) <- . delta(ql,b,ql) <- .

accept(S) <- initial(Q), accept_from(S,Q).

accept_from([],Q) <- final(Q).
accept_from([X[Xs],Q) <- delta(Q,X,NewQ), accept_from(Xs,NewQ) .




Recursive Programming: Automata (Graphs) (Contd.)

e A nondeterministic, stack, finite automaton (NDSFA):

accept(8) <- initial(Q), accept_from(S,Q,[]).

accept_from([],Q, [1) <- final(Q).
accept_from([X|Xs],Q,S) <- delta(Q,X,S,NewQ,NewS),
accept_from(Xs,NewQ,NewS) .

initial(q0) <- .
final(ql) <- .

delta(q0,X,Xs,q0, [X|Xs]) <- .
delta(q0,X,Xs,ql, [X|Xs]) <- .
delta(q0,X,Xs,ql,Xs) <- .

delta(ql,X, [X|Xs],ql,Xs) <- .

e What sequence does it recognize?




Recursive Programming: Towers of Hanoi

e Objective:

o Move tower of N disks from peg a to peg b, with the help of peg c.

e Rules:

¢ Only one disk can be moved at a time.
o A'larger disk can never be placed on top of a smaller disk.

a b [

b= 4l = | | = |




Recursive Programming: Towers of Hanoi (Contd.)

e We will call the main predicate hanoi moves (N,Moves)

e N is the number of disks and Moves the corresponding list of “moves”.
e Each move move (A, B) represents that the top disk in A should be moved to B.

e Example:

=1 1 S=1 L 1Ll Ll 4
A= e L= | | = |

IS represented by:

hanoi_moves( s(s(s(0))),
[ move(a,b), move(a,c), move(b,c), move(a,b),
move(c,a), move(c,b), move(a,b) 1)




Recursive Programming: Towers of Hanoi (Contd.)

e A general rule:

S N I Y B N B E N -
e \We capture this in a predicate |hanoi (N,Orig,Dest,Help,Moves)|where

“Moves contains the moves needed to move a tower of N disks from peg Orig to
peg Dest, with the help of peg Help.”

hanoi(s(0),0rig,Dest,_Help, [move(Orig, Dest)]) <- .
hanoi(s(N),0rig,Dest,Help,Moves) <-
hanoi(N,Orig,Help,Dest,Movesl),
hanoi(N,Help,Dest,0Orig,Moves2),
append (Moves1, [move(Orig, Dest) |Moves2] ,Moves) .

e And we simply call this predicate:

hanoi_moves(N,Moves) <-
hanoi(N,a,b,c,Moves).




Summary

e Pure logic programs allow purely declarative programming.

e Still, pure logic programming has full computational power.




